
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 7.1.2.3.7 which is part of the ENDC test suite in the ‘iwd-TTCN3-B2019-12_D19wk23’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.2.3.7 
Test Group:
\ENDC\7_1
ATS Version:
iwd-TTCN3-B2019-12_D19wk23
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
Qualcomm SDM855+SDX50
Verification Status:
PASS

4. Corrections required for ENDC TC 7.1.2.3.7
Change 4.1
	Function name
	

	Reason for change
	1. The UE will send two status reports at step 10. A Prose CR will be raised for this. Checks should be added as per TP2 of the TC to correctly check the time differences between Status Reports 
2. At step 24, v_TimingUL should be based on v_Timing DL1 and not v_TimingDL. t-PollRetransmit timer is used instead of t-StatusProhibit timer. The timer value should be passed as ms

3. AT step 39, v_TimingUL should be calculated based on v_TimingDL calculated at step 38, and v_timingDL should not be updated at step 39

	Summary of change
	1. Implemented checks to verify the time differnces between the Status Reports. Two consecutive Status reports hsould be apart by at least t-StatusProhibit timer value
2. Corrected the timer used to t-StatusProhibit and multiplied by 1000 to get the value in ms

3. Removed the step where v_timingDL is getting updated at step 39 just before v_timingUL is determined.

	TTCN module
	RLC_TC_Common_NR.ttcn

	MCC160 Comment
	1.  Accepted conditional to the approval of the accompanying prose CR at the next RAN5 meeting. 
2.  Accepted. 
3.  Accepted


Before Change
   function f_TC_7_1_2_3_7_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                            DRB_Identity p_NR_DRB_Id) runs on  NR_BASE_PTC

  {

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingUL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingDL2;

    var SubFrameTiming_Type v_TimingDL3;

    var integer v_RLC_SDUsize := 101; // With 2 octets of MAC header and 3 octets of RLC header an RLC PDU consists of 106 octets 
    var integer v_Duration;

    var SubFrameTiming_Type v_SubFrameTimingTa;

    var SubFrameTiming_Type v_SubFrameTimingTb;

    var SubFrameTiming_Type v_SubFrameTimingTc;

    var SubFrameTiming_Type v_SubFrameTimingTx;

    var template (value) NR_RLC_Status_NackSN18Bit_Type v_NR_NackSN18Bit_1;

    var template (value) NR_RLC_Status_NackSN18Bit_Type v_NR_NackSN18Bit_2;

    var template (value) NR_RLC_Status_NackListSN18Bit_Type v_NR_NackSN18Bit_List;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var float v_T_ReassemblyMax := f_NR_SetTimerToleranceMax (nr_Cell1, l2Timer, p_RLC_Rec.t_Reassembly);

    timer t_WatchDog;

…
…

…
    //@siclog "Step 10" siclog@

    // Check 1: Does the UE transmit a Status Report with NACK_SN=3, NACK_SN=7, ACK_SN=10?

    // Record time TC

    // Check 2: (TC - TB) = t-StatusProhibit?

    p_RLC_Rec.AM_RX_Next_Ack := 7; 

    v_NR_NackSN18Bit_List := { crs_NR_RLC_NACK_SN18_Simple(3)}; 

    v_SubFrameTimingTc := f_NR_RxSTATUS_PDU_GetTimingInfo(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);
    p_RLC_Rec.AM_RX_Next_Ack := 10;

    v_NR_NackSN18Bit_List := { crs_NR_RLC_NACK_SN18_Simple_E1(3), crs_NR_RLC_NACK_SN18_Simple(7) };

    v_SubFrameTimingTc := f_NR_RxSTATUS_PDU_GetTimingInfo(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    /*v_Duration := f_SubFrameTiming_Duration( v_SubFrameTimingTc_a, v_SubFrameTimingTc_b);

    if (f_NR_SubFrameTimingCheckDuration( nr_Cell1, l2Timer, v_Duration, float2int(p_RLC_Rec.t_StatusProhibit * 1000.0))) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10 Check 2");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10 Check 2");

    }*/
    //@siclog "Step 11" siclog@

    // The SS ignores scheduling requests unless otherwise specified and does not allocate any uplink grant.

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

…
…

…
    //@siclog "Step 24" siclog@

    // After 300 ms the SS transmits an AMD PDU with SN=11 and P=0, and an AMD PDU with SN=12 and P=1. A time spacing of 20 ms is applied.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 300);

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    v_TimingDL2 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, float2int(p_RLC_Rec.t_PollRetransmit/2.0)); // scheduling for step 26 !!
    p_RLC_Rec.AM_TX_Next := 11;

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_NoPoll, cs_TimingInfo_NR(v_TimingDL));

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_Poll, cs_TimingInfo_NR(v_TimingDL1));

    //@siclog "Step 26" siclog@

    // After t-Reassembly/2 after the transmission of the second AMD PDU of Step 24, the SS transmits an AMD PDU with SN=10 and P=0.

    // i.e. scheduled in such a way that step 25 is finished when the transmission is performed

    p_RLC_Rec.AM_TX_Next := 10;

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_NoPoll, cs_TimingInfo_NR(v_TimingDL2));

…
…

…
    //@siclog "Step 38" siclog@

    // After 300 ms the SS transmits an AMD PDU with SN=16 and P=1.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 300);

    p_RLC_Rec.AM_TX_Next := 16;

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_Poll, cs_TimingInfo_NR(v_TimingDL));

    //@siclog "Step 39" siclog@

    // 60 ms after step 38 the SS assigns an UL grant (UL grant allocation type 3) of size 112 bits.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1);

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 60);
    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(112, v_NR_UplinkBWP);

    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);

    //@siclog "Step 40" siclog@

    // Check: Does the UE transmit a Status Report with ACK_SN=20 and  NACK_SN: 13 including NACK Range 3 (SN 13, 14, 15) and NACK_SN=18 without NACK Range?

    p_RLC_Rec.AM_RX_Next_Ack := 20;

    v_NR_NackSN18Bit_1 := crs_NR_RLC_NACK_SN18_Range (13, 3);

    v_NR_NackSN18Bit_2 := crs_NR_RLC_NACK_SN18_Simple (18);

    v_NR_NackSN18Bit_List := { v_NR_NackSN18Bit_1, v_NR_NackSN18Bit_2 };

    v_SubFrameTimingTx := f_NR_RxSTATUS_PDU_GetTimingInfo(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

…
…

…
  }   
After Change
   function f_TC_7_1_2_3_7_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                            DRB_Identity p_NR_DRB_Id) runs on  NR_BASE_PTC

  {

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingUL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingDL2;

    var SubFrameTiming_Type v_TimingDL3;

    var integer v_RLC_SDUsize := 101; // With 2 octets of MAC header and 3 octets of RLC header an RLC PDU consists of 106 octets

    var integer v_Duration;

    var SubFrameTiming_Type v_SubFrameTimingTa;

    var SubFrameTiming_Type v_SubFrameTimingTb;

    var SubFrameTiming_Type v_SubFrameTimingTc;

    var SubFrameTiming_Type v_SubFrameTimingTc_a;

    var SubFrameTiming_Type v_SubFrameTimingTc_b;
    var SubFrameTiming_Type v_SubFrameTimingTx;

    var template (value) NR_RLC_Status_NackSN18Bit_Type v_NR_NackSN18Bit_1;

    var template (value) NR_RLC_Status_NackSN18Bit_Type v_NR_NackSN18Bit_2;

    var template (value) NR_RLC_Status_NackListSN18Bit_Type v_NR_NackSN18Bit_List;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var float v_T_ReassemblyMax := f_NR_SetTimerToleranceMax (nr_Cell1, l2Timer, p_RLC_Rec.t_Reassembly);

    timer t_WatchDog;

…
…

…
    //@siclog "Step 10" siclog@

    // Check 1: Does the UE transmit a Status Report with NACK_SN=3, NACK_SN=7, ACK_SN=10?

    // Record time TC

    // Check 2: (TC - TB) = t-StatusProhibit?

    p_RLC_Rec.AM_RX_Next_Ack := 7;

    v_NR_NackSN18Bit_List := { crs_NR_RLC_NACK_SN18_Simple(3)};

    v_SubFrameTimingTc_a := f_NR_RxSTATUS_PDU_GetTimingInfo(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    v_Duration := f_SubFrameTiming_Duration( v_SubFrameTimingTb, v_SubFrameTimingTc_a);

    if (f_NR_SubFrameTimingCheckDuration( nr_Cell1, l2Timer, v_Duration, float2int(p_RLC_Rec.t_StatusProhibit * 1000.0))) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10 Check 2");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10 Check 2a");

    }   
    p_RLC_Rec.AM_RX_Next_Ack := 10;

    v_NR_NackSN18Bit_List := { crs_NR_RLC_NACK_SN18_Simple_E1(3), crs_NR_RLC_NACK_SN18_Simple(7) }; 

    v_SubFrameTimingTc_b := f_NR_RxSTATUS_PDU_GetTimingInfo(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);   
    v_Duration := f_SubFrameTiming_Duration( v_SubFrameTimingTc_a, v_SubFrameTimingTc_b);

    if (f_NR_SubFrameTimingCheckDuration( nr_Cell1, l2Timer, v_Duration, float2int(p_RLC_Rec.t_StatusProhibit * 1000.0))) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10 Check 2");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10 Check 2b");

    }
    //@siclog "Step 11" siclog@

    // The SS ignores scheduling requests unless otherwise specified and does not allocate any uplink grant.

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

…
…

…
    //@siclog "Step 24" siclog@

    // After 300 ms the SS transmits an AMD PDU with SN=11 and P=0, and an AMD PDU with SN=12 and P=1. A time spacing of 20 ms is applied.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 300);

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    v_TimingDL2 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL1, float2int((p_RLC_Rec.t_Reassembly/2.0)*1000.0)); // scheduling for step 26 !! 
    p_RLC_Rec.AM_TX_Next := 11;

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_NoPoll, cs_TimingInfo_NR(v_TimingDL));

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_Poll, cs_TimingInfo_NR(v_TimingDL1));

    //@siclog "Step 26" siclog@

    // After t-Reassembly/2 after the transmission of the second AMD PDU of Step 24, the SS transmits an AMD PDU with SN=10 and P=0.

    // i.e. scheduled in such a way that step 25 is finished when the transmission is performed

    p_RLC_Rec.AM_TX_Next := 10;

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_NoPoll, cs_TimingInfo_NR(v_TimingDL2));

…
…

…
    //@siclog "Step 38" siclog@

    // After 300 ms the SS transmits an AMD PDU with SN=16 and P=1.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 300);

    p_RLC_Rec.AM_TX_Next := 16;

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_Poll, cs_TimingInfo_NR(v_TimingDL));

    //@siclog "Step 39" siclog@

    // 60 ms after step 38 the SS assigns an UL grant (UL grant allocation type 3) of size 112 bits.

    //v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // ( Removed
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 60);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(112, v_NR_UplinkBWP);

    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);

    //@siclog "Step 40" siclog@

    // Check: Does the UE transmit a Status Report with ACK_SN=20 and  NACK_SN: 13 including NACK Range 3 (SN 13, 14, 15) and NACK_SN=18 without NACK Range?

    p_RLC_Rec.AM_RX_Next_Ack := 20;

    v_NR_NackSN18Bit_1 := crs_NR_RLC_NACK_SN18_Range (13, 3);

    v_NR_NackSN18Bit_2 := crs_NR_RLC_NACK_SN18_Simple (18);

    v_NR_NackSN18Bit_List := { v_NR_NackSN18Bit_1, v_NR_NackSN18Bit_2 };

    v_SubFrameTimingTx := f_NR_RxSTATUS_PDU_GetTimingInfo(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

…
…

…
  }  

Change 4.2
	Function name
	f_NR_RLC_Get_AMD_FullSDU_TX

	Reason for change
	It is observed that p_RLC_Rec.AM_TX_Next value isn’t incremented between function calls.

	Summary of change
	Update the definition of function f_NR_RLC_Get_AMD_FullSDU_TX () to pass the parameter p_RLC_Rec by reference instead of by value.

	TTCN module
	NR_RLC_Common.ttcn

	MCC160 Comment
	Accepted. Implemented per R5s190317


Before Change
   function f_NR_RLC_Get_AMD_FullSDU_TX(NR_RLC_SS_State_Type p_RLC_Rec,

                                       B1_Type p_Poll := tsc_NR_P_NoPoll,

                                       NR_RLC_AMD_Data_Type p_Data) return template (value) NR_RLC_AMD_PDU_Type

  { // @sic R5s190179 sic@

    var template (value) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU := f_NR_RLC_Get_AMD_PDU_TX (p_RLC_Rec, p_Poll, -, -, p_Data);

    f_NR_RLC_SS_State_AM_TX_Next_Incr(p_RLC_Rec); // increment AM_TX_Next

    return v_NR_RLC_AMD_PDU;

  }  
After Change
  function f_NR_RLC_Get_AMD_FullSDU_TX(inout NR_RLC_SS_State_Type p_RLC_Rec,

                                       B1_Type p_Poll := tsc_NR_P_NoPoll,

                                       NR_RLC_AMD_Data_Type p_Data) return template (value) NR_RLC_AMD_PDU_Type

  { // @sic R5s190179 sic@

    var template (value) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU := f_NR_RLC_Get_AMD_PDU_TX (p_RLC_Rec, p_Poll, -, -, p_Data);

    f_NR_RLC_SS_State_AM_TX_Next_Incr(p_RLC_Rec); // increment AM_TX_Next

    return v_NR_RLC_AMD_PDU;

  } 

Change 4.3
	Function name
	f_NR_RLC_Get_AMD_FullSDU_RX

	Reason for change
	It is observed that p_RLC_Rec.AM_RX_Next value isn’t incremented between function calls.

	Summary of change
	Update the definition of function f_NR_RLC_Get_AMD_FullSDU_RX () to pass the parameter p_RLC_Rec by reference instead of by value.

	TTCN module
	NR_RLC_Common.ttcn

	MCC160 Comment
	Accepted. Implemented per R5s190317


Before Change
   function f_NR_RLC_Get_AMD_FullSDU_RX(NR_RLC_SS_State_Type p_RLC_Rec,

                                       template (present) B1_Type p_Poll := tsc_NR_P_NoPoll,

                                       NR_RLC_AMD_Data_Type p_Data) return template (present) NR_RLC_AMD_PDU_Type

  { // @sic R5s190179 sic@

    var template NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU := f_NR_RLC_Get_AMD_PDU_RX (p_RLC_Rec, p_Poll, -, -, p_Data);

    f_NR_RLC_SS_State_AM_RX_Next_Incr(p_RLC_Rec); // increment AM_RX_Next

    return v_NR_RLC_AMD_PDU;

  }  
After Change
  function f_NR_RLC_Get_AMD_FullSDU_RX(inout NR_RLC_SS_State_Type p_RLC_Rec,

                                       template (present) B1_Type p_Poll := tsc_NR_P_NoPoll,

                                       NR_RLC_AMD_Data_Type p_Data) return template (present) NR_RLC_AMD_PDU_Type

  { // @sic R5s190179 sic@

    var template NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU := f_NR_RLC_Get_AMD_PDU_RX (p_RLC_Rec, p_Poll, -, -, p_Data);

    f_NR_RLC_SS_State_AM_RX_Next_Incr(p_RLC_Rec); // increment AM_RX_Next

    return v_NR_RLC_AMD_PDU;

  } 

5. Execution Log Files

5.1 Qualcomm SDM855+SDX50
The Qualcomm SDM855+SDX50 UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on EN-DC band combination DC_5A_n261A.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_7_1_2_3_7_Log.html
· 

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

	R5s190457:   Supporting information for addition of EN-DC PDCP test case 7_1_2_3_7 in FR2 path
                          This archive comprises:

                        -  html and text format execution log files for Keysight test platform



